Data Visualization Through 3-D Graphs Using SAS® Graph Template Language (GTL)

Venu Perla, Ph.D.
Clinical Programmer, Emmes Corporation, Rockville, MD 20850
SAS Certified Base Programmer for SAS 9
SAS Certified Advanced Programmer for SAS 9
SAS Certified Clinical Trials Programmer Using SAS 9
SAS Certified Statistical Business Analyst Using SAS 9: Regression and Modeling

ABSTRACT

Certain types of data are better visualized through 3-D graphs. SAS Graph procedures available prior to SAS 9.2 version are not user friendly. These procedures are hard to practice and require quite a bit of time to implement. Since future of graphs in SAS is centered on Graph Template Language (GTL), the objective of this paper is to create 3-D graphs using SAS GTL. This paper also explains how initial SAS GTL code can be obtained, and how STATGRAPH template (aka GTL template) and SGRENDER procedure in the obtained code are modified with different OPTIONS and GTL STATEMENTS to create 3-D graphs for the beginners.

INTRODUCTION

Before SAS 9.2, graph creation was a tedious job. ODS (Output Delivery System) graphics software from SAS version 9.2 and above has provided an opportunity to create quick quality graphics using GTL. STATGRAPH templates can be created using TEMPLATE procedure in GTL. Features of GTL can support simplest as well as complex graphs. New features are being added in new versions of SAS. SAS 9.4 supports data skins for many different types of plots. More details about getting started with the GTL in SAS can be found in a recent publication by Matange (2013).

GTL based graph creation in SAS involves two-steps. Step 1: STATGRAPH template is created using GTL after defining the graph structure. Graphs are not generated at this stage. Step 2: SGRENDER procedure is employed to associate the data with the STATGRAPH template. Graph is created after execution of this step. This paper utilized SASHELP.HEART dataset to create following two types of 3D graphs in SAS:

1. 3D Surface Plot (SURFACEPLOTPARAM)
2. 3D Histogram (BIHISTOGRAM3DPARAM)

HEARTX dataset was created for this paper from SASHELP.HEART dataset. HEARTX dataset contained eight non-smoking, normal weighing, deceased male participants, reported to have high blood pressure and high cholesterol values (Table 1). HEARTX dataset was used for developing 3D graphs to understand the relationship between diastolic and systolic blood pressure, cholesterol levels and cause of death in the participants. To generate 3D graphs, data should be in (x,y,z) form. Where, z is the vertical axis. It is not feasible to manipulate data in plot statements. For this reason, it is required to generate x, y and z values in final format before using them in STATGRAPH template and SGRENDER procedure.

**HEARTX dataset for 3D graphs;
data heartx;
set sashelp.heart;
where (Sex='Male') and (BP_Status='High') and (Chol_Status='High') and (Weight_Status='Normal') and (Smoking=0) and (AgeAtDeath ^=.);
keep Status DeathCause Sex Diastolic Systolic Smoking AgeAtDeath Cholesterol Chol_Status BP_Status Weight_Status Smoking_Status;
run;
Data Visualization Through 3-D Graphs Using SAS® Graph Template Language (GTL), continued

GETTING INITIAL SAS GTL CODE FOR 3D GRAPH

This paper used ODS GRAPHICS DESIGNER (aka DESIGNER) tool of SAS 9.4 while generating initial graph. DESIGNER is an interactive GUI application available under TOOLS in the SAS. One can create a graph very quickly using this application without much GTL knowledge. Interestingly, DESIGNER also generates two-step code while creating a graph. Furthermore, one can save a generated graph template as a `.sgd` file for future use.

Steps for getting initial SAS GTL code for 3D Graph (Display 1-7):

1. SAS 9.4 > TOOLS >

2. ODS GRAPHICS DESIGNER >

```
title "Table 1. HEARTX Dataset.";
proc print data=heartx;
run;
```

Table 1. HEARTX Dataset.

<table>
<thead>
<tr>
<th>Obs</th>
<th>Status</th>
<th>DeathCause</th>
<th>Sex</th>
<th>Diastolic</th>
<th>Systolic</th>
<th>Smoking</th>
<th>AgeAtDeath</th>
<th>Cholesterol</th>
<th>Chol_Status</th>
<th>BP_Status</th>
<th>Weight_Status</th>
<th>Smoking_Status</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Dead</td>
<td>Cancer</td>
<td>Male</td>
<td>104</td>
<td>160</td>
<td>0</td>
<td>63</td>
<td>255</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>2</td>
<td>Dead</td>
<td>Cerebral Vascular Disease</td>
<td>Male</td>
<td>90</td>
<td>150</td>
<td>0</td>
<td>63</td>
<td>246</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>3</td>
<td>Dead</td>
<td>Coronary Heart Disease</td>
<td>Male</td>
<td>100</td>
<td>154</td>
<td>0</td>
<td>63</td>
<td>300</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>4</td>
<td>Dead</td>
<td>Coronary Heart Disease</td>
<td>Male</td>
<td>90</td>
<td>170</td>
<td>0</td>
<td>69</td>
<td>242</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>5</td>
<td>Dead</td>
<td>Coronary Heart Disease</td>
<td>Male</td>
<td>149</td>
<td>200</td>
<td>0</td>
<td>73</td>
<td>277</td>
<td>High</td>
<td>Non-smoker</td>
<td></td>
<td>Non-smoker</td>
</tr>
<tr>
<td>6</td>
<td>Dead</td>
<td>Unknown</td>
<td>Male</td>
<td>52</td>
<td>150</td>
<td>0</td>
<td>83</td>
<td>250</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>7</td>
<td>Dead</td>
<td>Coronary Heart Disease</td>
<td>Male</td>
<td>86</td>
<td>184</td>
<td>0</td>
<td>76</td>
<td>288</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
<tr>
<td>8</td>
<td>Dead</td>
<td>Coronary Heart Disease</td>
<td>Male</td>
<td>95</td>
<td>175</td>
<td>0</td>
<td>71</td>
<td>250</td>
<td>High</td>
<td>Normal</td>
<td>Non-smoker</td>
<td></td>
</tr>
</tbody>
</table>

Display 1. SAS 9.4 screenshot.
3. GRAPH GALLERY > CONTOUR PLOT (or any graph) >

Perhaps, 3D graphs (SURFACEPLOTPARAM and BIHISTOGRAM3DPARAM) are not available in GRAPH GALLERY. However, click on any graph which is close to 3D graph. Example: CONTOUR PLOT.
4. **ASSIGNING DATA > OK >**
Accept default settings while ASSIGNING DATA.

Display 4. **ASSIGN DATA pop-up window.**

5. Drag and drop required INSETS into CONTOUR GRAPH. Example: GRADIENT LEGEND.

Display 5. **INSETS and GRAPH windows.**

6. Now, an initial graph is generated with various options. This graph can be customized and saved as `.sgd` file for future reuse.
Display 6. Initial graph generated by ODS GRAPhICS DESIGNER.

Step 1

```
proc template;
define statgraph sgdesign:
dynamic WIDTH LENGTH DEPTH;
begingraph / designwidth=538 designheight=356;
  entrytitle align=center 'Type in your title...';
  entryfootnote align=left 'Type in your footnote...';
  layout lattice / rowdatarange=data columndatarange=data rowgutter=10
columngutter=10;
  layout overlay;
    contourplotparm x=_WIDTH y=_LENGTH z=_DEPTH / name='contour'
countourtype=GRADIENT colormodel=ThreeColorRamp reversecolormodel=true gridded=false;
    continuouslegend 'contour' / align=right valign=center location=outside;
    endlayout;
  endlayout;
endgraph;
end;
run;
```

Display 7. Graph CODE window.

Step 2

```
proc sgrender data=SASHELP.LAKE template=sgdesign;
dynamic WIDTH="WIDTH" LENGTH="LENGTH" DEPTH="DEPTH";
run;
```
8. Copy and paste the code in the SAS EDITOR. Modify the code for 3D graph(s) and run.

```
**Initial code from ODS Graphics Designer;
**Step 1;
proc template;
define statgraph sgdesign;
dynamic _WIDTH _LENGTH _DEPTH;
begingraph / designwidth=538 designheight=356;
    entrytitle halign=center 'Type in your title...';
    entryfootnote halign=left 'Type in your footnote...';
    layout lattice / rowdatarange=data
columndatarange=data
rowgutter=10
columngutter=10;
    layout overlay;
        contourplotparm x=_WIDTH
        y=_LENGTH
        z=_DEPTH /name='contour'
countourtype=GRADIENT
colormodel=ThreeColorRamp
reversecolormodel=true
gridded=false;
        continuouslegend 'contour' /
            halign=right
            valign=center
            location=outside;
    endlayout;
endlayout;
endgraph;
end;
run;
**Step 2;
proc sgrender data=SASHELP.LAKE template=sgdesign;
dynamic _WIDTH="WIDTH" _LENGTH="LENGTH" _DEPTH="DEPTH";
run;
```

Other options of DESIGNER can be further exploited to improve the graph appearance. Perhaps, any graph of choice can be generated using this technique.

### 3D SURFACE PLOT

The relationship between diastolic and systolic blood pressure, and cholesterol levels can be visualized better through 3D surface plot (SURFACEPLOTPARAM). MEANS procedure was employed to identify minimum and maximum axes values required for plotting grid lines on surface plot (Table 2). Smooth gridlines were generated by G3GRID procedure (Log 1 and Table 3). Lowest BY values in AXIS1 and AXIS2 statements in G3GRID procedure generate smooth grids in 3D surface plot.

```
title "Table 2. Minimum and Maximum Values for Axes.";
proc means data=heartx;
    var diastolic systolic cholesterol;
run;
```

### Table 2. Minimum and Maximum Values for Axes.

<table>
<thead>
<tr>
<th>Variable</th>
<th>N</th>
<th>Mean</th>
<th>Std Dev</th>
<th>Minimum</th>
<th>Maximum</th>
</tr>
</thead>
<tbody>
<tr>
<td>Diastolic</td>
<td>8</td>
<td>94.5250000</td>
<td>24.2306624</td>
<td>52.0000000</td>
<td>140.0000000</td>
</tr>
<tr>
<td>Systolic</td>
<td>8</td>
<td>169.1250000</td>
<td>17.1167211</td>
<td>150.0000000</td>
<td>200.0000000</td>
</tr>
<tr>
<td>Cholesterol</td>
<td>8</td>
<td>263.5000000</td>
<td>21.7781017</td>
<td>242.0000000</td>
<td>300.0000000</td>
</tr>
</tbody>
</table>
**Generation of Smooth Gridlines;**
proc g3grid data=heartx out=heart1;
   grid diastolic*systolic=cholesterol / spline
   axis1=50 to 140 by 1
   axis2=150 to 200 by 1;
run;

NOTE: There were 8 observations read from the data set WORK.HEARTX.
NOTE: The data set WORK.HEART1 has 4641 observations and 3 variables.

NOTE: PROCEDURE G3GRID used (Total process time):
   real time 0.02 seconds
   cpu time 0.00 seconds

Log 1. Log from PROCEDURE G3GRID.

proc sort data=heart1 out=heart2;
   by diastolic systolic;
run;

title "Table 3. Generation of Smooth Gridlines (Partial List).";
proc print data=heart2 (obs=10);
run;

<table>
<thead>
<tr>
<th>Obs</th>
<th>Diastolic</th>
<th>Systolic</th>
<th>Cholesterol</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>50</td>
<td>150</td>
<td>250.476</td>
</tr>
<tr>
<td>2</td>
<td>50</td>
<td>151</td>
<td>251.265</td>
</tr>
<tr>
<td>3</td>
<td>50</td>
<td>152</td>
<td>252.013</td>
</tr>
<tr>
<td>4</td>
<td>50</td>
<td>153</td>
<td>252.719</td>
</tr>
<tr>
<td>5</td>
<td>50</td>
<td>154</td>
<td>253.386</td>
</tr>
<tr>
<td>6</td>
<td>50</td>
<td>155</td>
<td>254.017</td>
</tr>
<tr>
<td>7</td>
<td>50</td>
<td>156</td>
<td>254.618</td>
</tr>
<tr>
<td>8</td>
<td>50</td>
<td>157</td>
<td>255.196</td>
</tr>
<tr>
<td>9</td>
<td>50</td>
<td>158</td>
<td>256.760</td>
</tr>
<tr>
<td>10</td>
<td>50</td>
<td>159</td>
<td>256.320</td>
</tr>
</tbody>
</table>

Code generated above by DESIGNER was modified and sorted dataset (HEART2) was used to generate colorful 3D surface graph (Figure 1). Note modifications made between LAYOUT and ENDLAYOUT statements in the code. For further improvement of 3D graph, refer various other options in the SAS documentation. In fact, most of the options are readily available in DESIGNER.

**3D Surface plot;**
**Step 1:**
proc template;
define statgraph sesug1;
begingraph;
   entrytitle halign=center 'Cholesterol: Diastolic vs Systolic';
   entryfootnote halign=center 'All the participants were male non-smokers, weighed normal and deceased.';
   entryfootnote halign=center 'All the participants had high blood pressure and high cholesterol.';
layout overlay3d/cube=false;
surfaceplotparm x=diastolic 
y=systolic 
z=cholesterol/name= "chol"
surfacetypename=fillgrid
surfacecolorgradient=cholesterol;
continuouslegend "chol"/
title="Cholesterol"
halign=right;
endlayout;
endgraph;
end;
run;

**Step 2;
proc sgrender data=heart2 template=sesug1;
run;

Figure 1. 3D SURFACE PLOT: Relationship between diastolic and systolic blood pressure, and cholesterol levels. Notice a hill on the surface plot that indicates an unusual surge in cholesterol levels in the participants.

3D HISTOGRAM

Similar to 3D Surface plot, the relationship between diastolic and systolic blood pressure levels, and number of deaths can be visualized better through 3D histogram (BIHISTOGRAM3DPARAM).

For execution of BIHISTOGRAM3DPARAM statement in 3D histogram template, data should be uniformly binned and sorted. For this reason, data ranges were divided into equal sized bins or intervals and observations were distributed into the bin combinations. This binning was performed on the HEARTX dataset using KDE procedure (Table 4). Execution of KDE procedure also generated kernel density plot for diastolic and systolic variables (Figure 2). Resulting binned dataset (HEART3) was used in SGRENDER procedure while creating the 3D histogram (Table 5). Alternatively, SUMMARY procedure can be used to generate binned dataset.
title "Table 4. Binning of Data with PROC KDE.";
proc kde data=heartx;
  bivar diastolic(ngrid=7) systolic(ngrid=6) /
    out=heart3(keep=value1 value2 count);
run;

Table 4. Binning of Data with PROC KDE.

<table>
<thead>
<tr>
<th>The KDE Procedure</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Inputs</strong></td>
</tr>
<tr>
<td>Data Set</td>
</tr>
<tr>
<td>Number of Observations Used</td>
</tr>
<tr>
<td>Variable 1</td>
</tr>
<tr>
<td>Variable 2</td>
</tr>
<tr>
<td>Bandwidth Method</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Controls</th>
<th>Diastolic</th>
<th>Systolic</th>
</tr>
</thead>
<tbody>
<tr>
<td>Grid Points</td>
<td>7</td>
<td>6</td>
</tr>
<tr>
<td>Lower Grid Limit</td>
<td>52</td>
<td>150</td>
</tr>
<tr>
<td>Upper Grid Limit</td>
<td>140</td>
<td>200</td>
</tr>
<tr>
<td>Bandwidth Multiplier</td>
<td>1</td>
<td>1</td>
</tr>
</tbody>
</table>

Figure 2. Kernel density plot generated by KDE procedure.
Similar to 3D Surface plot, code generated above by the DESIGNER was modified and HEART3 dataset was utilized to generate 3D histogram (Table 5 and Figure 3). Note modifications made between LAYOUT and ENDLAYOUT statements in the code.

```sas
title "Table 5. Binned Dataset: HEART3 (Partial List).";
proc print data=heart3;
run;

**3D Histogram;
**Step 1;
proc template;
define statgraph sesug2;
begingraph;
   entrytitle halign=center 'Number of Deaths: Diastolic vs Systolic';
   entryfootnote halign=left 'Note 1: Refer corresponding cholesterol data in Figure 1 (3D Surface plot).';
   entryfootnote halign=left 'Note 2: Refer HEARTX dataset (Table 1) for DEATH CAUSE.';
   layout overlay3d/cube=false zaxisopts=(griddisplay=on);
      bihistogram3dparm x=value1 y=value2 z=count/
         binaxis=true display=all;
   endlayout;
endgraph;
end;
run;

**Step 2;
**execute with binned dataset obtained with PROC KDE;
proc sgrender data= heart3 template=sesug2;
   label value1="Diastolic" value2="Systolic"
      count="Death Count";
run;
```

Table 5. Binned Dataset-HEART3 (Partial List).

<table>
<thead>
<tr>
<th>Obs</th>
<th>value1</th>
<th>value2</th>
<th>count</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>52.000</td>
<td>150</td>
<td>1</td>
</tr>
<tr>
<td>2</td>
<td>52.000</td>
<td>160</td>
<td>0</td>
</tr>
<tr>
<td>3</td>
<td>52.000</td>
<td>170</td>
<td>0</td>
</tr>
<tr>
<td>4</td>
<td>52.000</td>
<td>180</td>
<td>0</td>
</tr>
<tr>
<td>5</td>
<td>52.000</td>
<td>190</td>
<td>0</td>
</tr>
<tr>
<td>6</td>
<td>52.000</td>
<td>200</td>
<td>0</td>
</tr>
<tr>
<td>7</td>
<td>66.667</td>
<td>150</td>
<td>0</td>
</tr>
<tr>
<td>8</td>
<td>66.667</td>
<td>160</td>
<td>0</td>
</tr>
<tr>
<td>9</td>
<td>66.667</td>
<td>170</td>
<td>0</td>
</tr>
<tr>
<td>10</td>
<td>66.667</td>
<td>180</td>
<td>0</td>
</tr>
<tr>
<td>11</td>
<td>66.667</td>
<td>190</td>
<td>0</td>
</tr>
<tr>
<td>12</td>
<td>66.667</td>
<td>200</td>
<td>0</td>
</tr>
<tr>
<td>13</td>
<td>81.333</td>
<td>150</td>
<td>0</td>
</tr>
</tbody>
</table>
Figure 3. 3D HISTOGRAM: Relationship between diastolic and systolic blood pressure, and number of deaths.

CONCLUSION

Together, SAS GTL and ODS GRAPHIC DESIGNER made SAS graphs simple. The relationship between 3 variables can be better visualized through 3D SURFACE PLOT or 3D HISTOGRAM in SAS. Increasing demand for 3D graphs may change future of graphs in SAS.

REFERENCES


ACKNOWLEDGMENTS

I would like to thank the organizers for giving me an opportunity to present this paper at the SouthEast SAS Users Group Conference (SESUG 2016), October 16-18, 2016, Bethesda, MD, USA.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Venu Perla, Ph.D.
Emmes Corporation
401 N. Washington St, Suite 700
Rockville, MD 20850
Work Phone: 301 251 1161 Ext 2755
Fax: 800 781 9044
E-mail: venuperla@yahoo.com
Web: https://www.linkedin.com/pub/venu-perla/2a/700/468
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.